

```

10 REM ***** SLIPPERY *****
20 REM * ANGEL GARCIA DELGADO *
30 REM ***** G.T.S. *****
40 REM
50 MODE 0
60 INK 0,0:INK 1,26:INK 2,7:INK 3,18
70 PAPER 0:BORDER 0
80 ORIGIN 318,210
90 FOR a=0 TO 2*PI STEP PI/70
100 PLOT 0,0,2
110 DRAW 300*COS(a),150*SIN(a)
120 DRAW 0,30*SIN(a),3
130 NEXT a
140 PRINT CHR$(22)+CHR$(1);
150 LOCATE 7,12
160 PRINT "SLIPPERY"
170 PRINT CHR$(22)+CHR$(0);
180 INK 2,7,0:INK 3,0,18
190 FOR a=1 TO 300:SOUND 1,a,1,15:NEXT a
200 INK 2,18,7:INK 3,7,18
210 FOR a=300 TO 1 STEP -2:SOUND 1,a,1:SOUND 2,a*2,1:NEXT a
220 INK 2,7:INK 3,18
230 FOR a=1 TO 500:NEXT a
240 FOR a=0 TO 50:PRINT CHR$(7):FOR b=1 TO 50:NEXT b:NEXT a
250 MODE 1
260 GOTO 310
270 PRINT TAB(20-LEN(a$)/2):FOR a=1 TO LEN(a$)
280 b$=MID$(a$,a,1):PRINT b$;:IF b$<>" " THEN PRINT CHR$(7);
290 NEXT a
300 PRINT:PRINT:FOR a=1 TO 200:NEXT a:RETURN
310 REM *** INICIO DEL PROGRAMA ***
320 A$="<< SLIPPERY >>":GOSUB 270
330 PRINT:A$="AYUDA AL GUSANO SLIPPERY PARA QUE COJA":GOSUB 270
340 A$="LOS CAMELOS QUE HAY EN LA HABITACION":GOSUB 270
350 A$=" PERO TENIENDO MUCHO CUIDADO, PARA NO":GOSUB 270
360 A$="CHOCAR CONSIGO MISMO O CONTRA EL MURO":GOSUB 270
370 PRINT:A$="PARA DIRIGIRLO EMPLEA:":GOSUB 270
380 A$="Q. ARRIBA":GOSUB 270
390 A$="A. ABAJO":GOSUB 270
400 A$="O. IZQUIERDA":GOSUB 270
410 A$="P. DERECHA":GOSUB 270
420 PRINT:A$=" PULSA -S- PARA EMPEZAR":GOSUB 270
430 IF INKEY(60)<>0 THEN 430
440 GOSUB 750:P=0:V$="@@@"
450 INK 0,0:PAPER 0:BORDER 0:INK 1,6:INK 2,18,15:INK 3,12:PEN 1:MODE 1
460 WINDOW 1,40,3,25
470 X=10:Y=10:D=4
480 A$=STRING$(40,"[")
490 PEN 1:PRINT A$;:FOR A=1 TO 20
500 PEN 1:PRINT"[";:PEN 2:PRINT STRING$(38,"\\");:PEN 1:PRINT"[";

```

```

510 NEXT A:PRINT A$;
520 GOSUB 710
530 ENT 1,5,-1,1
540 IF INKEY(67)=0 THEN D=1
550 IF INKEY(69)=0 THEN D=2
560 IF INKEY(34)=0 THEN D=3
570 IF INKEY(27)=0 THEN D=4
580 X=X+(1 AND D=2)-(1 AND D=1):Y=Y+(1 AND D=4)-(1 AND D=3)
590 LOCATE Y,X:CALL 30000:C=PEEK(29999)
600 IF C=92 THEN SOUND 1,100,5,15,0,1:P=P+5:GOSUB 740
610 IF C=91 OR C=64 THEN GOSUB 640
620 LOCATE Y,X:PEN 3:PRINT"@ "
630 GOTO 540
640 REM ::: PIERDE VIDA :::
650 V$=MID$(V$,2,3)
660 FOR A=100 TO 1 STEP -2:SOUND 2,A,1,15,0,0,31:NEXT A
670 GOSUB 710
680 X=1+INT(RND*20):Y=1+INT(RND*39)
690 LOCATE Y,X:CALL 30000:IF PEEK(29999)=92 THEN RETURN
700 GOTO 680
710 PEN#1,1:LOCATE #1,1,1:PRINT#1,"PUNTOS:";USING "#####";P:PEN#1,3:LOCATE #1,3
0,1:PRINT #1,V$;" ";
720 IF LEN(V$)=0 THEN PEN#1,2:PRINT#1,"R.I.P.":GOTO 850
730 RETURN
740 PEN#1,1:LOCATE #1,8,1:PRINT#1,USING "#####";P:RETURN
750 REM ** CODE SCREEN$ **
760 DATA 205,96,187,50,47,117,201
770 RESTORE 760
780 FOR a=30000 TO 30006:READ b:POKE a,b:NEXT a
790 GOTO 800
800 SYMBOL AFTER 32
810 SYMBOL 64,60,102,223,191,191,255,126,60
820 SYMBOL 92,0,0,24,44,60,24,0,0
830 SYMBOL 91,0,239,239,239,0,254,254,254
840 RETURN
850 REM *** GAME OVER ***
860 FOR A=1 TO 20:FOR B=100 TO 200 STEP A:SOUND 1,B,1,15,0,0,31:NEXT B:NEXT A
870 INK 0,0:INK 1,26:INK 2,15,0:PAPER 0:BORDER 0:PEN 1:MODE 0
880 PEN 2:LOCATE 6,7:PRINT"GAME OVER"
890 PEN 1:LOCATE 4,12:PRINT"PUNTOS:";USING "#####";P
900 PEN 3:INK 3,15:LOCATE 5,17:PRINT "OTRA? (S/N)"
910 FOR A=1 TO 50:A$=INKEY$:NEXT A
920 PEN 1
930 IF INKEY(60)=0 THEN RUN 250
940 IF INKEY(46)=0 THEN CALL 0
950 GOTO 930

```